

Maple 17 では、エッセイの言語的な解析を行って採点するための新しいパッケージが追加されました。コンピュータがエッセイを採点する能力は、本質的に数学的です。コンピュータは、すでに手作業で採点済みの複数のエッセイを使用してエッセイ内のパターンを検索し、与えられたスコアに従って重み付けを行います。キーワード、文法、長さ、語彙などがスコアに大きく影響します。Maple のスコアリングモデルは、最大 20 個のアルゴリズムから選択できます。それぞれのアルゴリズムは、多数のプロパティを測定することで、新しいエッセイのスコアを予測するためのモデルを形成します。

EssayTools パッケージには、以下の機能があります。

- エッセイの採点
- 盗用の検出および類似点の比較
- 英単語の語形解析および文の簡約
- スペルチェック

採点コマンドは助言的な立場での使用に最も適しています。宿題や練習問題に対する学生の回答を見通すのに適しています。また、採点者が多くいる場合のダブルチェックにも有効です。たとえば、誰かが付けたスコアとコンピュータが付けたスコアが 1 ポイントまたは 2 ポイント以上異なっている箇所にフラグを設定し、別の人が再度採点することができます。

正確な予測を得るには、すでにスコアが付けられたエッセイを 100 以上用意します。多ければ多いほど正確になります。この概要においては、コマンドの形式およびこのような技術の潜在的な危険や制限を説明するために、データが不十分な例題を使用しています。

"Why is the sky blue?" という短答式の問題を考えてみましょう。いくつかの回答を収集して、配列に入れます。この配列を Answers とします。2 つ目の配列に、各回答の採点を設定します。この例では、2 つ目の配列を Scores とします。実際には、これらの回答と採点は ImportVector または ExcelTools:-Import コマンドのいずれかを使用して .csv ファイルから読み込むことができます。

```
> with(EssayTools);
[AppendToWordList, BinaryCosineCoefficient, BinaryDiceCoefficient,
  BinaryJaccardCoefficient, BuildScoreModel, CosineCoefficient, CountMisspellings,
  CountUseOfAllWords, CountUseOfEachWord, DetectPlagiarism, DiceCoefficient,
  IsAdjective, IsAdverb, IsConjunction, IsDefiniteArticle, IsIndefiniteArticle,
  IsInterjection, IsIntransitiveVerb, IsNominative, IsNoun, IsNounPhrase, IsPlural,
  IsPreposition, IsPronoun, IsTransitiveVerb, IsUsuallyParticipleVerb, IsVerb,
  JaccardCoefficient, Lemma, Misspellings, PartOfSpeech, QuadraticWeightedKappa,
  Reduce, Score, SetWordList, SimilarityScore, SpellCorrectWord, WordUse]
> Question := "Why is the sky blue?":
> Answers := Array(): Scores := Array():
> Answers(1) :=
```

"The blue color of the sky is due to Rayleigh scattering. As light moves through the atmosphere, most of the longer wavelengths pass straight through. Little of the red, orange and yellow light is affected by the air. However, much of the shorter wavelength light is absorbed by the gas molecules. The absorbed blue light is then radiated in different directions. It gets scattered all around the sky. Whichever direction you look, some of this scattered blue light reaches you. Since you see the blue light from everywhere overhead, the sky looks blue. ":

> Scores(1) := 10: #www.sciencemadesimple.com/sky_blue.html

> Answers(2) :=

"Sunlight reaches Earth's atmosphere and is scattered in all directions by all the gases and particles in the air. Blue light is scattered in all directions by the tiny molecules of air in Earth's atmosphere. Blue is scattered more than other colors because it travels as shorter, smaller waves. This is why we see a blue sky most of the time.":

> Scores(2) := 10: #spaceplace.nasa.gov/blue-sky

> Answers(3) :=

"A clear cloudless day-time sky is blue because molecules in the air scatter blue light from the sun more than they scatter red light.":

> Scores(3) := 8: #math.ucr.edu/home/baez/physics/General/BlueSky/blue_sky.html

> Answers(4) :=

"The blue color of the sky is due to Rayleigh scattering. As light moves through the atmosphere, most of the longer wavelengths pass straight through. Little of the red, orange and yellow light is affected by the air. However, much of the shorter wavelength light is absorbed by the gas molecules. The absorbed blue light is then radiated in different directions. It gets scattered all around the sky. Whichever direction you look, some of this scattered blue light reaches you. Since you see the blue light from everywhere overhead, the sky looks blue.":

> Scores(4) := 10: #www.sciencemadesimple.com/sky_blue.html

> Answers(5) :=

"Sunlight is a mix of all the colours of the rainbow. Light is a spectrum of light of different wavelengths, from the shorter blue to the longer red wavelengths. When sunlight enters the atmosphere, the molecules in the air are the right size to scatter the blue wavelengths of the sunlight, which sort of paints the sky blue.":

> Scores(5) := 7: #answers.yahoo.com/question/index?qid=20071031201452AAwbqie

> Answers(6) :=

"Blue light is scattered by the air molecules in the atmosphere (referred to as Rayleigh scattering). The blue wavelength is scattered more, because the scattering effect increases with the

```

inverse of the fourth power of the incident wavelength. ":
> Scores(6) := 7: #answers.yahoo.com/question/index?qid=
20071031201452AAwbqie
> Answers(7) := "The sky is blue because of a reflection of the sea.":
> Scores(7) := 0:
> Answers(8) := "The sky is blue because the sun is yellow and the
grass is green. Yellow and green make blue.":
> Scores(8) := 0:
> Answers(9) := "These days, the atmosphere is mostly nitrogen and
oxygen. Sunlight is made up of all the colors of the rainbow (as
well as many wavelengths we can't see); as it jostles through air
molecules, blue light is most efficiently reflected, so our eyes end
up experiencing a beautiful azure shade":
> Scores(9) := 10: #news.discovery.com/earth/no-seriously-why-is-the-
sky-blue.html

```

EssayTools パッケージから BuildScoreModel コマンドを使用してモデルを生成します。

```

> model := BuildScoreModel(Answers, Scores, questionText = Question);
                                model := [ Scoring Model ]
                                          [ 9 Responses ]

```

(2)

たとえば LibraryTools:-Save(model,"blue_sky.mla"); を呼び出して、このモデルを保存することができます。後から回答が増えた場合は、libname := libname, "blue_sky.mla"; をポイントして変数 model を使用することができます。

ここで、前は見られなかった回答を確認してみましょう。

```

> NewResponse :=
  "Blue and violet wavelengths, on the other hand, are absorbed by the
  gas molecules and scattered across the sky. Your eye sees these
  reflected wavelengths as blue.":
  #wonderopolis.org/wonder/why-is-the-sky-blue/
> Score(NewResponse,model);
                                7

```

(3)

その他の EssayTools コマンドを使用して、裏側で何が行われているかを把握することができます。まず、モデル回答の中で出現頻度の高い単語を調べます。

```

> WordUse(Answers, showcount, mincount = 3);

```

```
[ "the" = 56, "of" = 24, "blue" = 24, "is" = 24, "light" = 19, "sky" = 11, "in" = 9, "scattered" = 9, "air" = 8, "as" = 7, "atmosphere" = 7, "all" = 7, "and" = 7, "molecules" = 7, "by" = 7, "wavelengths" = 6, "a" = 6, "you" = 6, "through" = 5, "sunlight" = 5, "because" = 5, "to" = 5, "see" = 4, "it" = 4, "shorter" = 4, "from" = 4, "most" = 4, "wavelength" = 4, "yellow" = 4, "absorbed" = 4, "directions" = 4, "red" = 4, "reaches" = 3, "scattering" = 3, "this" = 3, "longer" = 3, "more" = 3, "scatter" = 3, "rayleigh" = 3, "different" = 3 ] (4)
```

これらの頻出単語を使用して文章を作ってみます。

```
[ > Score("The is of blue light sky in scattered air.", model); (5)
      5
```

```
[ > Score("Blue light sky scattered air molecules.", model); (6)
      6
```

このモデルは "molecules" という単語を過度に重視しています。この単語は常に高スコアの回答で使用され、低スコアの回答には見られません。これは、モデルを構築するためには大量のデータが必要であることを強調しています。

簡約手法を使用すると、重要でない単語を取り除き、類似した意味を持つ単語を合体し、回答を意味のある小さなフレーズに分割することで、エッセイを簡略化することができます。

```
[ > ModifiedResponse := Reduce(NewResponse);
  ModifiedResponse := ["blue violet wavelength hand be absorbed gas molecule scatter sky", (7)
    "eye see reflected wavelength blue"]
```

このプロセスにおける重要なツールは、lemmas という単語の使用です。

```
[ > Lemma("is");
      "be" (8)
```

```
[ > Lemma("are");
      "be" (9)
```

```
[ > Lemma("flew");
      "fly" (10)
```

```
[ > Lemma("flies");
      "fly" (11)
```

文を区切る箇所を見分けるには、それぞれの従属節が通常 1 つ以上の名詞と動詞を持ち、接続詞で分割されている必要があります。

```
[ > TwoIdeas := "the sky is blue because it scatters sunlight."; (12)
  TwoIdeas := "the sky is blue because it scatters sunlight."
```

```
> map(PartOfSpeech, StringTools:-Words(TwoIdeas));
[["Definite Article", "Adverb"], ["Noun", "Verb (usually participle)"],
  ["Verb (usually participle)", ["Noun", "Adjective", "Verb (usually participle)"],
  ["Conjunction"], ["Pronoun", "Noun"], FAIL, ["Noun"]] (13)
```

品詞情報がない単語もあります。上述の例では、"scatters" は認識されず、FAIL が返されています。Reduce コマンドは、Lemma コマンドを使用して元となる単語を取得します。

```
> Lemma("scatters");
"scatter" (14)
```

```
> PartOfSpeech(%);
["Verb (transitive)", "Noun", "Verb (usually participle)"] (15)
```

```
> Reduce(TwoIdeas);
["sky be blue", "sky scatter sunlight"] (16)
```

エッセイを採点する別の手法は、モデルセットから最も類似したエッセイを検索することです。類似性を計算するにはさまざまな方法があります。

```
> SimilarityScore(NewResponse, Answers);
[0.1818181818, 0.2200000000, 0.1621621622, 0.1818181818, 0.1489361702,
  0.1666666667, 0.1428571429, 0.1666666667, 0.1851851852] (17)
```

デフォルトの類似性基準 BinaryJaccardCoefficient では、2 番目のエッセイが最もスコアが高く 0.22 のスコアが付けられています。それは最も類似しているという意味になります。

```
> SimilarityScore(NewResponse, Answers, methods = [CosineCoefficient]);
;
[0.5687043153, 0.5075459214, 0.4181210050, 0.5687043153, 0.5408420486,
  0.5304958918, 0.3108809143, 0.5017452060, 0.4717820977] (18)
```

```
> SimilarityScore(NewResponse, Answers, methods = [DiceCoefficient]);
[0.3652694611, 0.4285714286, 0.4166666667, 0.3652694611, 0.3820224719,
  0.4958677686, 0.2692307692, 0.5000000000, 0.4504504505] (19)
```

また、CosineCoefficient 基準では回答 1 と 4 が選ばれ、DiceCoefficient 基準では回答 8 が選ばれています。これらの基準およびそれに対応するバイナリは、それぞれ異なる方法で類似性を測定します。

類似性スコアが特定の最小点を越えている場合、そのエッセイの一部または全体が他のエッセイをコピーしたものである可能性が高くなります。

```
> DetectPlagiarism(Answers);
[[1, 4, 1.], [7, 8, 0.3529411765]] (20)
```

DetectPlagiarism コマンドは、指定されたすべてのエッセイを比較し、最小類似性スコアを超えるものにフラグを設定します。この場合、エッセイ 1 とエッセイ 4 の類似性スコアは 1 であるため、明らかにコピーです。エッセイ 7 と 8 はどちらも短く、まったく同じ 5 つの単語で始まっているため、コピーの可能性があると指摘されています。

参照

[EssayTools](#)